// SPDX-License-Identifier: MIT
pragma solidity >= 0.8.7  < 0.9.0;

//import "../utils/Context.sol";
/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}


//import "@openzeppelin/contracts/access/Ownable.sol";
/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _setOwner(_msgSender());
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
        _;
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _setOwner(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _setOwner(newOwner);
    }

    function _setOwner(address newOwner) private {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

//import "./IERC721Receiver.sol";
/**
 * @title ERC721 token receiver interface
 * @dev Interface for any contract that wants to support safeTransfers
 * from ERC721 asset contracts.
 */
interface IERC721Receiver {
    /**
     * @dev Whenever an {IERC721} `tokenId` token is transferred to this contract via {IERC721-safeTransferFrom}
     * by `operator` from `from`, this function is called.
     *
     * It must return its Solidity selector to confirm the token transfer.
     * If any other value is returned or the interface is not implemented by the recipient, the transfer will be reverted.
     *
     * The selector can be obtained in Solidity with `IERC721.onERC721Received.selector`.
     */
    function onERC721Received(
        address operator,
        address from,
        uint256 tokenId,
        bytes calldata data
    ) external returns (bytes4);
}


//import "../../utils/introspection/IERC165.sol";
/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}


//import "./IERC721.sol";
/**
 * @dev Required interface of an ERC721 compliant contract.
 */
interface IERC721 is IERC165 {
    /**
     * @dev Emitted when `tokenId` token is transferred from `from` to `to`.
     */
    event Transfer(address indexed from, address indexed to, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables `approved` to manage the `tokenId` token.
     */
    event Approval(address indexed owner, address indexed approved, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables or disables (`approved`) `operator` to manage all of its assets.
     */
    event ApprovalForAll(address indexed owner, address indexed operator, bool approved);

    /**
     * @dev Returns the number of tokens in ``owner``'s account.
     */
    function balanceOf(address owner) external view returns (uint256 balance);

    /**
     * @dev Returns the owner of the `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function ownerOf(uint256 tokenId) external view returns (address owner);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be have been allowed to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Transfers `tokenId` token from `from` to `to`.
     *
     * WARNING: Usage of this method is discouraged, use {safeTransferFrom} whenever possible.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Gives permission to `to` to transfer `tokenId` token to another account.
     * The approval is cleared when the token is transferred.
     *
     * Only a single account can be approved at a time, so approving the zero address clears previous approvals.
     *
     * Requirements:
     *
     * - The caller must own the token or be an approved operator.
     * - `tokenId` must exist.
     *
     * Emits an {Approval} event.
     */
    function approve(address to, uint256 tokenId) external;

    /**
     * @dev Returns the account approved for `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function getApproved(uint256 tokenId) external view returns (address operator);

    /**
     * @dev Approve or remove `operator` as an operator for the caller.
     * Operators can call {transferFrom} or {safeTransferFrom} for any token owned by the caller.
     *
     * Requirements:
     *
     * - The `operator` cannot be the caller.
     *
     * Emits an {ApprovalForAll} event.
     */
    function setApprovalForAll(address operator, bool _approved) external;

    /**
     * @dev Returns if the `operator` is allowed to manage all of the assets of `owner`.
     *
     * See {setApprovalForAll}
     */
    function isApprovedForAll(address owner, address operator) external view returns (bool);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes calldata data
    ) external;
}


//import "./extensions/IERC721Metadata.sol";
/**
 * @title ERC-721 Non-Fungible Token Standard, optional metadata extension
 * @dev See https://eips.ethereum.org/EIPS/eip-721
 */
interface IERC721Metadata is IERC721 {
    /**
     * @dev Returns the token collection name.
     */
    function name() external view returns (string memory);

    /**
     * @dev Returns the token collection symbol.
     */
    function symbol() external view returns (string memory);

    /**
     * @dev Returns the Uniform Resource Identifier (URI) for `tokenId` token.
     */
    function tokenURI(uint256 tokenId) external view returns (string memory);
}


//import "../../utils/Address.sol";
/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize, which returns 0 for contracts in
        // construction, since the code is only stored at the end of the
        // constructor execution.

        uint256 size;
        assembly {
            size := extcodesize(account)
        }
        return size > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(isContract(target), "Address: delegate call to non-contract");

        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verifies that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly

                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}


//import "../../utils/Strings.sol";
/**
 * @dev String operations.
 */
library Strings {
    bytes16 private constant _HEX_SYMBOLS = "0123456789abcdef";

    /**
     * @dev Converts a `uint256` to its ASCII `string` decimal representation.
     */
    function toString(uint256 value) internal pure returns (string memory) {
        // Inspired by OraclizeAPI's implementation - MIT licence
        // https://github.com/oraclize/ethereum-api/blob/b42146b063c7d6ee1358846c198246239e9360e8/oraclizeAPI_0.4.25.sol

        if (value == 0) {
            return "0";
        }
        uint256 temp = value;
        uint256 digits;
        while (temp != 0) {
            digits++;
            temp /= 10;
        }
        bytes memory buffer = new bytes(digits);
        while (value != 0) {
            digits -= 1;
            buffer[digits] = bytes1(uint8(48 + uint256(value % 10)));
            value /= 10;
        }
        return string(buffer);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation.
     */
    function toHexString(uint256 value) internal pure returns (string memory) {
        if (value == 0) {
            return "0x00";
        }
        uint256 temp = value;
        uint256 length = 0;
        while (temp != 0) {
            length++;
            temp >>= 8;
        }
        return toHexString(value, length);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation with fixed length.
     */
    function toHexString(uint256 value, uint256 length) internal pure returns (string memory) {
        bytes memory buffer = new bytes(2 * length + 2);
        buffer[0] = "0";
        buffer[1] = "x";
        for (uint256 i = 2 * length + 1; i > 1; --i) {
            buffer[i] = _HEX_SYMBOLS[value & 0xf];
            value >>= 4;
        }
        require(value == 0, "Strings: hex length insufficient");
        return string(buffer);
    }
}


// import "../../utils/introspection/ERC165.sol";
/**
 * @dev Implementation of the {IERC165} interface.
 *
 * Contracts that want to implement ERC165 should inherit from this contract and override {supportsInterface} to check
 * for the additional interface id that will be supported. For example:
 *
 * ```solidity
 * function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
 *     return interfaceId == type(MyInterface).interfaceId || super.supportsInterface(interfaceId);
 * }
 * ```
 *
 * Alternatively, {ERC165Storage} provides an easier to use but more expensive implementation.
 */
abstract contract ERC165 is IERC165 {
    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IERC165).interfaceId;
    }
}


//import "@openzeppelin/contracts/token/ERC721/ERC721.sol";
/**
 * @dev Implementation of https://eips.ethereum.org/EIPS/eip-721[ERC721] Non-Fungible Token Standard, including
 * the Metadata extension, but not including the Enumerable extension, which is available separately as
 * {ERC721Enumerable}.
 */
contract ERC721 is Context, ERC165, IERC721, IERC721Metadata {
    using Address for address;
    using Strings for uint256;

    // Token name
    string private _name;

    // Token symbol
    string private _symbol;

    // Mapping from token ID to owner address
    mapping(uint256 => address) private _owners;

    // Mapping owner address to token count
    mapping(address => uint256) private _balances;

    // Mapping from token ID to approved address
    mapping(uint256 => address) private _tokenApprovals;

    // Mapping from owner to operator approvals
    mapping(address => mapping(address => bool)) private _operatorApprovals;

    /**
     * @dev Initializes the contract by setting a `name` and a `symbol` to the token collection.
     */
    constructor(string memory name_, string memory symbol_) {
        _name = name_;
        _symbol = symbol_;
    }

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override(ERC165, IERC165) returns (bool) {
        return
            interfaceId == type(IERC721).interfaceId ||
            interfaceId == type(IERC721Metadata).interfaceId ||
            super.supportsInterface(interfaceId);
    }

    /**
     * @dev See {IERC721-balanceOf}.
     */
    function balanceOf(address owner) public view virtual override returns (uint256) {
        require(owner != address(0), "ERC721: balance query for the zero address");
        return _balances[owner];
    }

    /**
     * @dev See {IERC721-ownerOf}.
     */
    function ownerOf(uint256 tokenId) public view virtual override returns (address) {
        address owner = _owners[tokenId];
        require(owner != address(0), "ERC721: owner query for nonexistent token");
        return owner;
    }

    /**
     * @dev See {IERC721Metadata-name}.
     */
    function name() public view virtual override returns (string memory) {
        return _name;
    }

    /**
     * @dev See {IERC721Metadata-symbol}.
     */
    function symbol() public view virtual override returns (string memory) {
        return _symbol;
    }

    /**
     * @dev See {IERC721Metadata-tokenURI}.
     */
    function tokenURI(uint256 tokenId) public view virtual override returns (string memory) {
        require(_exists(tokenId), "ERC721Metadata: URI query for nonexistent token");

        string memory baseURI = _baseURI();
        return bytes(baseURI).length > 0 ? string(abi.encodePacked(baseURI, tokenId.toString())) : "";
    }

    /**
     * @dev Base URI for computing {tokenURI}. If set, the resulting URI for each
     * token will be the concatenation of the `baseURI` and the `tokenId`. Empty
     * by default, can be overriden in child contracts.
     */
    function _baseURI() internal view virtual returns (string memory) {
        return "";
    }

    /**
     * @dev See {IERC721-approve}.
     */
    function approve(address to, uint256 tokenId) public virtual override {
        address owner = ERC721.ownerOf(tokenId);
        require(to != owner, "ERC721: approval to current owner");

        require(
            _msgSender() == owner || isApprovedForAll(owner, _msgSender()),
            "ERC721: approve caller is not owner nor approved for all"
        );

        _approve(to, tokenId);
    }

    /**
     * @dev See {IERC721-getApproved}.
     */
    function getApproved(uint256 tokenId) public view virtual override returns (address) {
        require(_exists(tokenId), "ERC721: approved query for nonexistent token");

        return _tokenApprovals[tokenId];
    }

    /**
     * @dev See {IERC721-setApprovalForAll}.
     */
    function setApprovalForAll(address operator, bool approved) public virtual override {
        require(operator != _msgSender(), "ERC721: approve to caller");

        _operatorApprovals[_msgSender()][operator] = approved;
        emit ApprovalForAll(_msgSender(), operator, approved);
    }

    /**
     * @dev See {IERC721-isApprovedForAll}.
     */
    function isApprovedForAll(address owner, address operator) public view virtual override returns (bool) {
        return _operatorApprovals[owner][operator];
    }

    /**
     * @dev See {IERC721-transferFrom}.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) public virtual override {
        //solhint-disable-next-line max-line-length
        require(_isApprovedOrOwner(_msgSender(), tokenId), "ERC721: transfer caller is not owner nor approved");

        _transfer(from, to, tokenId);
    }

    /**
     * @dev See {IERC721-safeTransferFrom}.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) public virtual override {
        safeTransferFrom(from, to, tokenId, "");
    }

    /**
     * @dev See {IERC721-safeTransferFrom}.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes memory _data
    ) public virtual override {
        require(_isApprovedOrOwner(_msgSender(), tokenId), "ERC721: transfer caller is not owner nor approved");
        _safeTransfer(from, to, tokenId, _data);
    }

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * `_data` is additional data, it has no specified format and it is sent in call to `to`.
     *
     * This internal function is equivalent to {safeTransferFrom}, and can be used to e.g.
     * implement alternative mechanisms to perform token transfer, such as signature-based.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function _safeTransfer(
        address from,
        address to,
        uint256 tokenId,
        bytes memory _data
    ) internal virtual {
        _transfer(from, to, tokenId);
        require(_checkOnERC721Received(from, to, tokenId, _data), "ERC721: transfer to non ERC721Receiver implementer");
    }

    /**
     * @dev Returns whether `tokenId` exists.
     *
     * Tokens can be managed by their owner or approved accounts via {approve} or {setApprovalForAll}.
     *
     * Tokens start existing when they are minted (`_mint`),
     * and stop existing when they are burned (`_burn`).
     */
    function _exists(uint256 tokenId) internal view virtual returns (bool) {
        return _owners[tokenId] != address(0);
    }

    /**
     * @dev Returns whether `spender` is allowed to manage `tokenId`.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function _isApprovedOrOwner(address spender, uint256 tokenId) internal view virtual returns (bool) {
        require(_exists(tokenId), "ERC721: operator query for nonexistent token");
        address owner = ERC721.ownerOf(tokenId);
        return (spender == owner || getApproved(tokenId) == spender || isApprovedForAll(owner, spender));
    }

    /**
     * @dev Safely mints `tokenId` and transfers it to `to`.
     *
     * Requirements:
     *
     * - `tokenId` must not exist.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function _safeMint(address to, uint256 tokenId) internal virtual {
        _safeMint(to, tokenId, "");
    }

    /**
     * @dev Same as {xref-ERC721-_safeMint-address-uint256-}[`_safeMint`], with an additional `data` parameter which is
     * forwarded in {IERC721Receiver-onERC721Received} to contract recipients.
     */
    function _safeMint(
        address to,
        uint256 tokenId,
        bytes memory _data
    ) internal virtual {
        _mint(to, tokenId);
        require(
            _checkOnERC721Received(address(0), to, tokenId, _data),
            "ERC721: transfer to non ERC721Receiver implementer"
        );
    }

    /**
     * @dev Mints `tokenId` and transfers it to `to`.
     *
     * WARNING: Usage of this method is discouraged, use {_safeMint} whenever possible
     *
     * Requirements:
     *
     * - `tokenId` must not exist.
     * - `to` cannot be the zero address.
     *
     * Emits a {Transfer} event.
     */
    function _mint(address to, uint256 tokenId) internal virtual {
        require(to != address(0), "ERC721: mint to the zero address");
        require(!_exists(tokenId), "ERC721: token already minted");

        _beforeTokenTransfer(address(0), to, tokenId);

        _balances[to] += 1;
        _owners[tokenId] = to;

        emit Transfer(address(0), to, tokenId);
    }

    /**
     * @dev Destroys `tokenId`.
     * The approval is cleared when the token is burned.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     *
     * Emits a {Transfer} event.
     */
    function _burn(uint256 tokenId) internal virtual {
        address owner = ERC721.ownerOf(tokenId);

        _beforeTokenTransfer(owner, address(0), tokenId);

        // Clear approvals
        _approve(address(0), tokenId);

        _balances[owner] -= 1;
        delete _owners[tokenId];

        emit Transfer(owner, address(0), tokenId);
    }

    /**
     * @dev Transfers `tokenId` from `from` to `to`.
     *  As opposed to {transferFrom}, this imposes no restrictions on msg.sender.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     *
     * Emits a {Transfer} event.
     */
    function _transfer(
        address from,
        address to,
        uint256 tokenId
    ) internal virtual {
        require(ERC721.ownerOf(tokenId) == from, "ERC721: transfer of token that is not own");
        require(to != address(0), "ERC721: transfer to the zero address");

        _beforeTokenTransfer(from, to, tokenId);

        // Clear approvals from the previous owner
        _approve(address(0), tokenId);

        _balances[from] -= 1;
        _balances[to] += 1;
        _owners[tokenId] = to;

        emit Transfer(from, to, tokenId);
    }

    /**
     * @dev Approve `to` to operate on `tokenId`
     *
     * Emits a {Approval} event.
     */
    function _approve(address to, uint256 tokenId) internal virtual {
        _tokenApprovals[tokenId] = to;
        emit Approval(ERC721.ownerOf(tokenId), to, tokenId);
    }

    /**
     * @dev Internal function to invoke {IERC721Receiver-onERC721Received} on a target address.
     * The call is not executed if the target address is not a contract.
     *
     * @param from address representing the previous owner of the given token ID
     * @param to target address that will receive the tokens
     * @param tokenId uint256 ID of the token to be transferred
     * @param _data bytes optional data to send along with the call
     * @return bool whether the call correctly returned the expected magic value
     */
    function _checkOnERC721Received(
        address from,
        address to,
        uint256 tokenId,
        bytes memory _data
    ) private returns (bool) {
        if (to.isContract()) {
            try IERC721Receiver(to).onERC721Received(_msgSender(), from, tokenId, _data) returns (bytes4 retval) {
                return retval == IERC721Receiver.onERC721Received.selector;
            } catch (bytes memory reason) {
                if (reason.length == 0) {
                    revert("ERC721: transfer to non ERC721Receiver implementer");
                } else {
                    assembly {
                        revert(add(32, reason), mload(reason))
                    }
                }
            }
        } else {
            return true;
        }
    }

    /**
     * @dev Hook that is called before any token transfer. This includes minting
     * and burning.
     *
     * Calling conditions:
     *
     * - When `from` and `to` are both non-zero, ``from``'s `tokenId` will be
     * transferred to `to`.
     * - When `from` is zero, `tokenId` will be minted for `to`.
     * - When `to` is zero, ``from``'s `tokenId` will be burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(
        address from,
        address to,
        uint256 tokenId
    ) internal virtual {}
}


//import "./IPresaleInfo.sol";
//-----------------------------------------------------------------------
// 事前販売情報：インターフェイス
//-----------------------------------------------------------------------
interface IPresaleInfo {
    //----------------------------------------
    // 登録確認されているか？
    //----------------------------------------
    function isRegistered( address user ) external view returns (bool);
}


//import "./LibString.sol";
library LibString {
    function toString(uint256 value) internal pure returns (string memory) {

        if (value == 0) {
            return "0";
        }
        uint256 temp = value;
        uint256 digits;
        while (temp != 0) {
            digits++;
            temp /= 10;
        }
        bytes memory buffer = new bytes(digits);
        while (value != 0) {
            digits -= 1;
            buffer[digits] = bytes1(uint8(48 + uint256(value % 10)));
            value /= 10;
        }
        return string(buffer);
    }

}


//-------------------
// トークン
//-------------------
contract Token is Ownable, ERC721{
    //----------------------------------------------
    // Events
    //----------------------------------------------
    event TokenStake( address indexed holder, uint256 indexed tokenId );
    event TokenUnstake( address indexed holder, uint256 indexed tokenId );

    //----------------------------------------------
    // 定数
    //----------------------------------------------
    string constant private TOKEN_NAME = "Comical Girl";
    string constant private TOKEN_SYMBOL = "CG";

    uint256 constant private PUBLIC_SALE_MAX_SHOT = 7;
    uint256 constant private PRE_SALE_MAX_SHOT = 2;
    uint256 constant private PRE_SALE_LIMIT = 2;
 
    uint256 constant private ID_PHASE_GA  = 0;
    uint256 constant private ID_PHASE_1ST = 1;
    uint256 constant private ID_PHASE_2ND = 2;
    uint256 constant private ID_PHASE_3RD = 3;
    uint256 constant private ID_PHASE_MAX = 4;

    //----------------------------------------------
    // 管理データ
    //----------------------------------------------
    // 売り上げを引き出せる人
    address private _accountant;

    // 排出管理
    uint256[ID_PHASE_MAX] private _arr_id_offset;
    uint256[ID_PHASE_MAX] private _arr_num_mintable;
    uint256[ID_PHASE_MAX] private _arr_num_minted;

    // metadata参照先
    string[ID_PHASE_MAX] private _arr_base_url;
    mapping( uint256 => string ) private _repaired_urls;

    // 販売関連
    uint256 private _sale_phase;

    uint256 private _sale_price;
    uint256 private _sale_start;
    uint256 private _sale_end;
    bool private _sale_suspended;

    uint256 private _presale_price;
    uint256 private _presale_start;
    uint256 private _presale_end;
    IPresaleInfo[] private _presale_infos;
    mapping( address => uint256 ) private _address_to_presale_minted_1st;
    mapping( address => uint256 ) private _address_to_presale_minted_2nd;
    mapping( address => uint256 ) private _address_to_presale_minted_3rd;
    bool private _presale_suspended;

    // ステーク関連
    bool private _stake_suspended;
    bool private _unstake_suspended;
    mapping( uint256 => address ) private _token_to_stake_holder;

    //--------------------------------------------
    // コンストラクタ
    //--------------------------------------------
    constructor() Ownable() ERC721( TOKEN_NAME, TOKEN_SYMBOL ){
        // お金を引き出せる人
        _accountant = 0xda1B1849EeF51f1F5EfF49dA1220ebb89a5ef2FF;   // 本番

        // id_offset
        _arr_id_offset[ID_PHASE_GA] = 1;
        _arr_id_offset[ID_PHASE_1ST] = 113;
        _arr_id_offset[ID_PHASE_2ND] = 1668;
        _arr_id_offset[ID_PHASE_3RD] = 4223;

        // num_mintable
        _arr_num_mintable[ID_PHASE_GA] = 112;
        _arr_num_mintable[ID_PHASE_1ST] = 1555;
        _arr_num_mintable[ID_PHASE_2ND] = 2555;
        _arr_num_mintable[ID_PHASE_3RD] = 3555;

        // ステーク関連（リリース時はオフにしておく）
        _stake_suspended = true;
        _unstake_suspended = true;
    }

    //---------------------------------------------------
    // [public] 一般販売が利用可能か？
    //---------------------------------------------------
    function isSaleAvailable() public view returns (bool) {
        // フェーズが無効
        if( _sale_phase < ID_PHASE_1ST || _sale_phase > ID_PHASE_3RD ){
            return( false );
        }

        // 一時停止中
        if( _sale_suspended ){
            return( false );
        }

        // 開始前
        if( _sale_start > block.timestamp ){
            return( false );
        }

        // 終了：_sale_endが０の場合は無期限
        if( _sale_end != 0 && _sale_end <= block.timestamp ){
            return( false );
        }

        // 利用可能
        return( true );
    }

    //--------------------------------------------
    // [external] 一般販売
    //--------------------------------------------    
    function mintTokens( uint256 num ) external payable {
        // 購入可能か？
        require( isSaleAvailable(), "sale: not available" );

        // 試行回数は有効か？
        require( num > 0 && num <= PUBLIC_SALE_MAX_SHOT, "sale: invalid num" );

        // 残りはあるか？
        require( _arr_num_mintable[_sale_phase] >= (_arr_num_minted[_sale_phase]+num), "sale: remaining not enough" );

        // 入金額は有効か？
        uint256 amount = _sale_price * num;
        require( amount <= msg.value , "sale: insufficient value" );
        
        //--------------------------
        // ここまできたらチェック完了
        //--------------------------

        for( uint256 i=0; i<num; i++ ){
            _safeMint( msg.sender, _arr_num_minted[_sale_phase] + _arr_id_offset[_sale_phase] );
            _arr_num_minted[_sale_phase]++;
        }
    }

    //---------------------------------------------------
    // [public] 事前販売が利用可能か？
    //---------------------------------------------------
    function isPresaleAvailable() public view returns (bool) {
        // フェーズが無効
        if( _sale_phase < ID_PHASE_1ST || _sale_phase > ID_PHASE_3RD ){
            return( false );
        }

        // 一時停止中
        if( _presale_suspended ){
            return( false );
        }

        // 開始前
        if( _presale_start > block.timestamp ){
            return( false );
        }

        // 終了：_presale_endが０の場合は無期限
        if( _presale_end != 0 && _presale_end <= block.timestamp ){
            return( false );
        }

        // 利用可能
        return( true );
    }

    //---------------------------------------------------
    // [public] 事前登録済みのユーザーか？
    //---------------------------------------------------
    function isRegisteredUser( address user ) public view returns (bool) {
        for( uint256 i=0; i<_presale_infos.length; i++ ){
            if( _presale_infos[i].isRegistered( user ) ){
                return( true );
            }
        }

        return( false );
    }

    //--------------------------------------------
    // [external] 事前販売
    //--------------------------------------------    
    function mintTokensForPresale( uint256 num ) external payable {
        // 購入可能か？
        require( isPresaleAvailable(), "pre-sale: not available" );

        // 事前登録に登録しているか？
        require( isRegisteredUser(msg.sender), "pre-sale: not registered" );

        // 試行回数は有効か？
        require( num > 0 && num <= PRE_SALE_MAX_SHOT, "pre-sale: invalid num" );

        // 発行上限未満か？
        uint256 minted = PRE_SALE_LIMIT;
        if( _sale_phase == ID_PHASE_1ST){ minted = _address_to_presale_minted_1st[msg.sender]; }
        else if( _sale_phase == ID_PHASE_2ND){ minted = _address_to_presale_minted_2nd[msg.sender]; }
        else if( _sale_phase == ID_PHASE_3RD){ minted = _address_to_presale_minted_3rd[msg.sender]; }

        require( PRE_SALE_LIMIT >= (minted+num), "pre-sale: limitation" );

        // 残りはあるか？
        require( _arr_num_mintable[_sale_phase] >= (_arr_num_minted[_sale_phase]+num), "pre-sale: remaining not enough" );

        // 入金額は有効か？
        uint256 amount = _presale_price * num;
        require( amount <= msg.value , "pre-sale: insufficient value" );
        
        //--------------------------
        // ここまできたらチェック完了
        //--------------------------

        for( uint256 i=0; i<num; i++ ){
            _safeMint( msg.sender, _arr_num_minted[_sale_phase] + _arr_id_offset[_sale_phase] );
            _arr_num_minted[_sale_phase]++;
        }

        if( _sale_phase == ID_PHASE_1ST){ _address_to_presale_minted_1st[msg.sender] += num; }
        else if( _sale_phase == ID_PHASE_2ND){ _address_to_presale_minted_2nd[msg.sender] += num; }
        else if( _sale_phase == ID_PHASE_3RD){ _address_to_presale_minted_3rd[msg.sender] += num; }
    }

    //--------------------------------------------
    // [external/onlyOwner] トークンのgiveaway
    //--------------------------------------------    
    function giveawayTokens( address[] calldata users ) external onlyOwner {
        // 発行上限未満か？
        require( users.length > 0 && _arr_num_mintable[ID_PHASE_GA] >= (_arr_num_minted[ID_PHASE_GA]+users.length), "giveaway: invalid length" );

        // 用心
        for( uint256 i=0; i<users.length; i++ ){
            require( users[i] != address(0), "giveaway: invalid user" );
        }

        //--------------------------
        // ここまできたらチェック完了
        //--------------------------

        for( uint256 i=0; i<users.length; i++ ){
            _safeMint( users[i], _arr_num_minted[ID_PHASE_GA] + _arr_id_offset[ID_PHASE_GA] );
            _arr_num_minted[ID_PHASE_GA]++;
        }
    }

    //--------------------------------------------------------
    // [external] 引き出せるアカウントの確認
    //--------------------------------------------------------
    function accountant() external view returns (address) {
        return( _accountant );
    }

    //--------------------------------------------------------
    // [external/onlyOwner] 引き出せるアカウントの設定
    //--------------------------------------------------------
    function setaAccountant( address target ) external onlyOwner {
        _accountant = target;
    }

    //--------------------------------------------------------
    // [external] 各種確認
    //--------------------------------------------------------
    function idOffsetAt( uint256 id ) external view returns (uint256) { return( _arr_id_offset[id] ); }
    function numMintableAt( uint256 id ) external view returns (uint256) { return( _arr_num_mintable[id] ); }
    function numMintedAt( uint256 id ) external view returns (uint256) { return( _arr_num_minted[id] ); }

    //--------------------------------------------------------
    // [external] 発行可能数の確認
    //--------------------------------------------------------
    function totalMintable() external view returns (uint256) {
        uint256 mintable = 0;
        for( uint256 i=0; i<ID_PHASE_MAX; i++ ){
            mintable += _arr_num_mintable[i];
        }
        return( mintable );
    }

    //--------------------------------------------------------
    // [external] 発行数の確認
    //--------------------------------------------------------
    function totalMinted() external view returns (uint256) {
        uint256 minted = 0;
        for( uint256 i=0; i<ID_PHASE_MAX; i++ ){
            minted += _arr_num_minted[i];
        }
        return( minted );
    }

    //--------------------------------------------------------
    // [external] _base_url の確認
    //--------------------------------------------------------
    function baseUrlAt( uint256 at ) external view returns (string memory) {
        return( _arr_base_url[at] );
    }

    //--------------------------------------------------------
    // [external/onlyOwner] _base_url の設定
    //--------------------------------------------------------
    function setBaseUrlAt( uint256 at, string calldata url ) external onlyOwner {
        _arr_base_url[at] = url;
    }

    //---------------------------------------------------
    // [public] 修復されたURLの確認
    //---------------------------------------------------
    function repairedUrl( uint256 tokenId ) public view returns (string memory) {
        require( _exists( tokenId ), "nonexistent token" );

        return( _repaired_urls[tokenId] );
    }

    //---------------------------------------------------
    // [external/onlyOwner] メタデータのURLの修復
    //---------------------------------------------------
    function repairUrl( uint256 tokenId, string calldata url ) external onlyOwner {
        require( _exists( tokenId ), "nonexistent token" );

        _repaired_urls[tokenId] = url;
    }

    //----------------------------------------------
    // [public] tokenURI
    //----------------------------------------------
    function tokenURI( uint256 tokenId ) public view override returns (string memory) {
        require( _exists( tokenId ), "nonexistent token" );

        // 修復データがあれば
        string memory url = repairedUrl( tokenId );
        if( bytes(url).length > 0 ){
            return( url );
        }

        uint256 at = ID_PHASE_MAX;
        for( uint256 i=0; i<ID_PHASE_MAX; i++ ){
            if( tokenId >= _arr_id_offset[i] && tokenId < (_arr_id_offset[i]+_arr_num_mintable[i]) ){
                at = i;
                break;
            }
        }

        require( at < ID_PHASE_MAX, "invalid phase" );

        // ここまできたらリリース時のメタを返す
        string memory strId = LibString.toString( tokenId );
        return( string( abi.encodePacked( _arr_base_url[at], strId ) ) );
    }

    //---------------------------------------------------
    // [external/onlyOwner] 販売設定
    //---------------------------------------------------
    function setSale( uint256 phase, uint256 saleP, uint256 saleF, uint256 saleT, uint256 presaleP, uint256 presaleF, uint256 presaleT, address[] calldata arrInfo ) external onlyOwner {
        _sale_phase = phase;

        _sale_price = saleP;
        _sale_start = saleF;
        _sale_end = saleT;

        _presale_price = presaleP;
        _presale_start = presaleF;
        _presale_end = presaleT;

        delete _presale_infos;
        for( uint256 i=0; i<arrInfo.length; i++ ){
            _presale_infos.push( IPresaleInfo( arrInfo[i] ) );
        }
    }

    //---------------------------------------------------
    // [external] 設定確認
    //---------------------------------------------------
    function salePhase() external view returns (uint256) { return(_sale_phase); }
    function salePrice() external view returns (uint256) { return(_sale_price); }
    function saleStart() external view returns (uint256) { return( _sale_start ); }
    function saleEnd() external view returns (uint256) { return( _sale_end ); }
    function presalePrice() external view returns (uint256) { return(_presale_price); }
    function presaleStart() external view returns (uint256) { return( _presale_start ); }
    function presaleEnd() external view returns (uint256) { return( _presale_end ); }
    function presaleInfoAt( uint256 at ) external view returns (address) {
        if( at < _presale_infos.length ){
            return( address(_presale_infos[at]) );
        }
        return( address(0) );
    }
    function presaleMintedAt( uint256 at, address user ) external view returns (uint256) {
        if( at == ID_PHASE_1ST){ return( _address_to_presale_minted_1st[user] ); }
        if( at == ID_PHASE_2ND){ return( _address_to_presale_minted_2nd[user] ); }
        if( at == ID_PHASE_3RD){ return( _address_to_presale_minted_3rd[user] ); }
        return( 0 );
    }

    //---------------------------------------------------
    // [external] 情報の一括取得（フロント向け）
    //---------------------------------------------------
    function saleInfo() external view returns (uint256[12] memory) {
        require( _sale_phase >= ID_PHASE_1ST && _sale_phase <= ID_PHASE_3RD, "invalid phase" );

        uint256[12] memory info;

        // 状況
        info[0] = _sale_phase;
        info[1] = _arr_id_offset[_sale_phase];
        info[2] = _arr_num_mintable[_sale_phase];
        info[3] = _arr_num_minted[_sale_phase];

        // セール情報
        if( isSaleAvailable() ){
            info[4] = 1;
        }
        info[5] = _sale_price;
        info[6] = _sale_start;
        info[7] = _sale_end;

        // プレセール情報
        if( isPresaleAvailable() ){
            info[8] = 1;
        }
        info[9] = _presale_price;
        info[10] = _presale_start;
        info[11] = _presale_end;

        return( info );
    }

    //---------------------------------------------------
    // [external/onlyOwner] 販売停止
    //---------------------------------------------------
    function suspendSale( bool flag ) external onlyOwner { _sale_suspended = flag; }
    function suspendPresale( bool flag ) external onlyOwner { _presale_suspended = flag; }

    //---------------------------------------------------
    // [external] 販売状況の確認
    //---------------------------------------------------
    function saleSuspended() external view returns (bool) { return( _sale_suspended ); }
    function presaleSuspended() external view returns (bool) { return( _presale_suspended ); }

    //---------------------------------------------------
    // [external/onlyOwner] ステークキングの停止
    //---------------------------------------------------
    function suspendStake( bool flag ) external onlyOwner { _stake_suspended = flag; }
    function suspendUnstake( bool flag ) external onlyOwner { _unstake_suspended = flag; }

    //---------------------------------------------------
    // [external] ステーキング状況の確認
    //---------------------------------------------------
    function stakeSuspended() external view returns (bool) { return( _stake_suspended ); }
    function unstakeSuspended() external view returns (bool) { return( _unstake_suspended ); }

    //--------------------------------------------------------
    // [external] NFTのステーク
    //--------------------------------------------------------
    function stakeTokens( uint256[] calldata tokenIds ) external {
        require( !_stake_suspended, "stake: not available" );

        // 確認
        for( uint256 i=0; i<tokenIds.length; i++ ){
            require( _exists(tokenIds[i]), "stake: nonexistent token" );
            require( ownerOf(tokenIds[i]) == msg.sender, "stake: not owner" );

            for( uint256 j=i+1; j<tokenIds.length; j++ ){
                require( tokenIds[i] != tokenIds[j], "stake: duplicated token id" );
            }
        }

        // オーナーに預け入れ
        for( uint256 i=0; i<tokenIds.length; i++ ){
            safeTransferFrom( msg.sender, owner(), tokenIds[i] );
            _token_to_stake_holder[tokenIds[i]] = msg.sender;
            
            emit TokenStake( msg.sender, tokenIds[i] );
        }
    }

    //--------------------------------------------------------
    // [external] NFTのアンステーク
    //--------------------------------------------------------
    function unstakeTokens( uint256[] calldata tokenIds ) external {
        require( !_unstake_suspended, "unstake: not available" );

        // 確認
        for( uint256 i=0; i<tokenIds.length; i++ ){
            require( _exists(tokenIds[i]), "unstake: nonexistent token" );
            require( _token_to_stake_holder[tokenIds[i]] == msg.sender, "unstake: not holder" );
 
             for( uint256 j=i+1; j<tokenIds.length; j++ ){
                require( tokenIds[i] != tokenIds[j], "unstake: duplicated token id" );
            }
       }

        // オーナーから払い戻し
        for( uint256 i=0; i<tokenIds.length; i++ ){
            safeTransferFrom( owner(), msg.sender, tokenIds[i] );
            delete _token_to_stake_holder[tokenIds[i]];

            emit TokenUnstake( msg.sender, tokenIds[i] );
        }
    }

    //--------------------------------------------------------
    // [external] トークンをステークしているアドレスの確認
    //--------------------------------------------------------
    function checkStakeHolder( uint256 tokenId ) external view returns (address) {
        require( _exists(tokenId), "nonexistent token" );

        return( _token_to_stake_holder[tokenId] );
    }

    //--------------------------------------------------------
    // [external] 残高の確認
    //--------------------------------------------------------
    function checkBalance() external view returns (uint256) {
        return( address(this).balance );
    }

    //--------------------------------------------------------
    // [external/onlyAccountant] 引き出し
    //--------------------------------------------------------
    function withdraw( uint256 amount ) external {
        require( msg.sender == _accountant, "not accountant" );
        require( amount <= address(this).balance, "insufficient balance" );

        address payable target = payable( msg.sender );
        target.transfer( amount );
    }

}